Мордовских Е.П.
методист по информатизации управления образования Администрации Катайского района

преподаватель информатики ГОУ СПО «Катайский педагогический техникум»
Урок представлен на «Фестиваль медиауроков-2010»
Тема: Геометрические фракталы

Дисциплина: Компьютерная графика
Комбинированное занятие (для студентов 4 курса со специализацией – информатика)

Задачи:
Образовательные:

· ввести понятие «фрактал»;

· познакомить учащихся с математическими моделями построения некоторых геометрических фракталов;

· закрепить навыки работы с языком программирования TPascal (в среде Delphi) по темам «графика», «процедуры и функции» и «рекурсия»;

Развивающие:

· развивать алгоритмическое мышление;

· развивать умение анализировать, сравнивать, и делать выводы;

Воспитательные:

· воспитывать культуру умственного труда;
· прививать интерес к предмету.
Метод обучения: частично-поисковый.
Оборудование:

1. компьютеры;
2. проектор или интерактивная доска.
Литература:

1. Гарднер М. От мозаик Пенроуза к надежным шифрам. 1993 г. 418 с.

2. Мандельброт Б. Фрактальная геометрия природы. 2002 г. 656 с.
3. Морозов А.Д. Введение в теорию фракталов. 2002 г. 160 с.

4. Федер Е. Фракталы. 1991 г. 260 с.
5. www.ait.ustu.ru/AIT/uch/Fractal
6. www.rechka.ru
7. fract.narod.ru
Продолжительность занятия – 90 минут (пара).
Ход урока
1 слайд

Сравнительно давно в математике возник образ объекта, более объемистого, но, тем не менее, сходного с линией. Некоторым ученым было трудно примириться с понятием линии, не имеющей ширины, поэтому постепенно ими стали изучаться геометрические формы и структуры, имеющие дробную пространственную размерность. На смену непрерывным кривым, обладающим всеми своими производными, пришли ломаные или очень изрезанные кривые. Ярким примером такой кривой является траектория броуновской частицы. Так в науке возникло понятие фрактала.
2 слайд

Фракталами называются геометрические объекты: линии, поверхности, пространственные тела, имеющие сильно изрезанную форму и обладающие свойством само подобия. Слово фрактал произошло от латинского слова fractus и переводится как дробный, ломаный.
3 слайд

Фракталы бывают 3 видов:

· геометрические фракталы, в их основе лежит некая математическая модель, которую можно выразить геометрически; это самая распространенная группа, когда говорят о фракталах, обычно имеют ввиду именно геометрические фракталы;
· алгебраические фракталы, в их основе лежит математическая модель нелинейных динамических систем; самый яркий пример – графическое представление комплексных чисел;
· системы итерируемых функций, графически они выражаются как геометрические фракталы, но имеют другую математическую модель построения;
· стохастические – это рандомизированные фракталы, каждый раз имеют разное изображение, поскольку модель зависит от случайных величин; пример – плазма.
Самоподобие как основная характеристика фрактала означает, что он более или менее единообразно устроен в широком диапазоне масштабов. Так, при увеличении маленькие фрагменты фрактала получаются очень похожими на большие. В идеальном случае такое самоподобие приводит к тому, что фрактальный объект оказывается инвариантным относительно растяжений, т.е. ему, как говорят, присуща дилатационная симметрия. Она предполагает неизменность основных геометрических особенностей фрактала при изменении масштаба.
Давайте рассмотрим на примерах.
1. Ковер Серпинского (4 слайд)

Рецепт его создания состоит в следующем. Вначале берется квадрат с длиной стороны, равной единице. Затем каждая из сторон квадрата делится на три равные части, а весь квадрат, соответственно, на девять одинаковых квадратиков со стороной, равной 1/3. Из полученной фигуры вырезается центральный квадрат. Затем такой же процедуре подвергается каждый из 8 оставшихся квадратиков и т. д.

Давайте попробуем это реализовать в Delphi.
В основной программе вводим исходные данные и вызываем процедуру рисования фрактальной фигуры.

var k:integer;

    x,y,p:real;

begin

  k:=5;


//количество итераций
  p:=729;


//размер квадрата
  x:=clientwidth shr 1-p/2;
//начальная координата по оси x
  y:=x;


//начальная координата по оси y
  Draw(x,y,p,k); 
//вызов рекурсивной процедуры
end.
Как же будет выглядеть сама рекурсивная процедура?
procedure draw(x,y,p:real;k:integer);

var p0:real;

begin

  p0:=p/3;
//длина стороны квадрата уменьшается в 3 раза
  if (k=0) or (p0<=1) then exit; {выход из процедуры если нарисовали все итерации либо если кончилась пиксельная сетка}
  k:=k-1;

//уменьшаем количество итераций
  x:=round(x);
//округляем координаты для красоты при прорисовке
  y:=round(y); 
  bar(x,y,x+p,y+p,$ff0000);
//рисование прямоугольника
  bar(x+p0,y+p0,x+2*p0,y+2*p0,$ffffff);

  draw(x,y,p0,k);


//рекурсивные вызовы процедуры
  draw(x+p0,y,p0,k);

  draw(x+2*p0,y,p0,k);

  draw(x,y+p0,p0,k);

  draw(x+2*p0,y+p0,p0,k);

  draw(x,y+2*p0,p0,k);

  draw(x+p0,y+2*p0,p0,k);

  draw(x+2*p0,y+2*p0,p0,k);

end;

2. Салфетка Серпинского (5 слайд)

Идея салфетки Серпинского подобна ковру. Берется треугольник со стороной, равной единице. Затем каждая из его сторон делится пополам, полученные 3 точки соединяют линиями. Таким образом треугольник оказывается разделенным на 4 треугольника. Центральный вырезается, а остальные три подвергаются той же процедуре и т.д.

Программу рисования салфетки Серпинского вы реализуете дома. Она подобна ковру, но есть отличие в программной части: на delphi функции, которая бы рисовала треугольник, поэтому вам придется прорисовывать стороны вручную. Как это сделать, зная начальные координаты отрезка x0 y0,  длину отрезка R и угол поворота a?

В раностороннем треугольнике угол поворота 60°. Длину отрезка и начальные координаты задаете самостоятельно в зависимости от размера формы, на которой все выводите.
3. Шестиугольник Серпинского (6 слайд)
На нулевой итерации шестиугольник Серпинского представляет собой обыкновенный шестиугольник. На следующей итерации этот шестиугольник убирается и вместо него рисуется 6 шестиугольников с центрами в вершинах исходной фигуры и с радиусом, уменьшенным в 3 раза. На следующей итерации эти 6 шестиугольников убираются и вместо каждого опять рисуется 6 фигур, таким образом будет уже 36 фигур. Этот процесс можно продолжать до бесконечности.

Отличие программной части в том, что не нужно прорисовывать каждую итерацию, а только последнюю. Давайте попробуем реализовать процедуру рисования шестиугольника Серпинского.
procedure draw(x,y,p:real;k:integer);

var p0,a,x1,y1,x2,y2:real;

    i:integer;

begin

  p0:=p/3; {длина стороны шестиугольника уменьшается в 3 раза}
  if (k=0) or (p<1) then //если это последняя итерация, то
    begin

      a:=0;



      x1:=x+p*cos(a);
//находим координаты первой точки 
      y1:=y+p*sin(a);
//шестиугольника по оси Х и У
      for i:=1 to 6 do
//затем в цикле 
        begin

          a:=a+pi/3;

          x2:=x+p*cos(a); //находим координаты остальных точек
          y2:=y+p*sin(a);

          line0(x1,y1,x2,y2);
//соединяем их линиями
          x1:=x2;

          y1:=y2;

        end;

      exit;

//и выходим из процедуры
    end;

  k:=k-1;

//иначе уменьшаем количество итераций на 1
  a:=0;

  for i:=1 to 6 do
//в цикле
    begin

      x1:=x+p*cos(a); //находим координаты точек шестиугольника
      y1:=y+p*sin(a);

      draw(x1,y1,p0,k); {для каждой точки шестиугольника вызываем рекурсивную процедуру} 
      a:=a+pi/3;

    end;

end;

4. Снежинка Коха (7 слайд)
Для построения снежинки Коха выполним следующие операции. Рассмотрим в качестве нулевой итерации равносторонний треугольник. Затем каждую из сторон этого треугольника разделим на три равные части, уберем среднюю часть и в середине достроим равносторонний треугольник так, как изображено на рисунке. На следующем шаге такой же процедуре деления на три равные части и достраивания равностороннего треугольника подвергается каждая из сторон новой фигуры, и так до бесконечности. В результате возникает симметричная, похожая на снежинку, бесконечно изломанная кривая, которая представляет собой самоподобное множество, называемое снежинкой Коха.

5. Кривая Госпера (8 слайд)

Инициатором для кривой Госпера является отрезок единичной длины, а генератор из 7 отрезков показан на рисунке. Пунктиром показана треугольная решетка, служащая своеобразной образующей для этого генератора. Следующие три шага процесса построения показаны на рисунке.

6. Дакон Хартера-Хейтуэя (9 слайд)

На нулевой итерации это отрезок, на второй 2, на третей 4 и т.д. Схема такая: каждый из отрезков прямой на следующем шаге заменяется на два отрезка, образующих боковые стороны равнобедренного прямоугольного треугольника, для которого исходный отрезок являлся бы гипотенузой. В результате отрезок как бы прогибается под прямым углом. Направление прогиба чередуется. Первый отрезок прогибается вправо (по ходу движения слева направо), второй — влево, третий — опять вправо и т.д.

А сейчас вы попробуете самостоятельно программно реализовать один из этих фракталов. Они примерно одинаковы по сложности, но длина кода будет разной. Давайте разделимся на 3 варианта:
1 вариант – снежинка Коха

2 вариант – кривая Госпера

3 вариант – дракон Хартера-Хейтуэя
При выполнении помните, что для этих фракталов применяется тот же принцип, что и для шестиугольника Серпинского – прорисовываем фигуры только на последней итерации.
1) Снежинка Коха строится из трех одинаковых элементов, поэтому в основной программе вызывайте рекурсивную процедуру последовательно 3 раза.
Процедура рисования снежинки Коха:
procedure draw(x,y,p,a:real;k:integer);

var x1,y1,x2,y2,x3,y3,p0,a1:real;

begin

  if k=0 then

    begin

      x1:=x+p*cos(a); y1:=y+p*sin(a);

      line(x,y,x1,y1);

      exit;

    end;

  k:=k-1;

  p0:=p/3;

  x1:=x+p0*cos(a); y1:=y+p0*sin(a);

  x2:=x+2*p0*cos(a); y2:=y+2*p0*sin(a);

  Draw(x,y,p0,a,k);

  Draw(x2,y2,p0,a,k);

  a1:=a-pi/3;

  Draw(x1,y1,p0,a1,k);

  x3:=x1+p0*cos(a1);   y3:=y1+p0*sin(a1);

  a1:=a1+2*pi/3;

  Draw(x3,y3,p0,a1,k);

end;
2) У кривой Госпера самая длинная процедура, потому что в основной программе целых 7 рекурсивных вызова, но я дам вам подсказку для построения математической модели – при рекурсивном вызове на каждой новой итерации поворачивайте фигуру на 19°.

Процедура рисования кривой Госпера:
procedure draw(x,y,p,a:real;k:integer);

var x1,y1,p0:real;

begin

  p0:=p/sqrt(7);

  if (k=0) or (p0<1) then

    begin

      x1:=x+p*cos(a); y1:=y+p*sin(a);

      line(x,y,x1,y1);

      exit;

    end;

  k:=k-1;

  a:=a-19*pi/180;

  draw(x,y,p0,a,k);

  x1:=x+p0*cos(a); y1:=y+p0*sin(a);

  a:=a+pi/3;

  x1:=x1+p0*cos(a); y1:=y1+p0*sin(a);

  draw(x1,y1,p0,a+pi,k);

  a:=a+2*pi/3;

  x1:=x1+p0*cos(a); y1:=y1+p0*sin(a);

  draw(x1,y1,p0,a+pi,k);

  a:=a-pi/3;

  draw(x1,y1,p0,a,k);

  x1:=x1+p0*cos(a); y1:=y1+p0*sin(a);

  a:=a-2*pi/3;

  draw(x1,y1,p0,a,k);

  x1:=x1+p0*cos(a); y1:=y1+p0*sin(a);

  draw(x1,y1,p0,a,k);

  x1:=x1+p0*cos(a); y1:=y1+p0*sin(a);

  a:=a-pi/3;

  x1:=x1+p0*cos(a); y1:=y1+p0*sin(a);

  draw(x1,y1,p0,a+pi,k);

end;

3) Процедура рисования дракона Хартера-Хейтуэя:
procedure draw(x,y,p,a:real;k:integer);

var x1,y1,p0:real;

begin

  p0:=p/sqrt(2);

  if k=0 then

    begin

      x1:=x+p*cos(a);

      y1:=y+p*sin(a);

      line(x,y,x1,y1);

      exit;

    end;

  k:=k-1;

  draw(x,y,p0,a-pi/4,k);

  x1:=x+p*cos(a);

  y1:=y+p*sin(a);

  draw(x1,y1,p0,a-3* pi/4,k);

end;
7. Терагон (10 слайд)
Для того чтобы сделать терагон, сначала нужно снежинку Коха заполнить кривыми Пиано.

Инициатором будет семизвездный генератор.

После того как кривые будет нарисованы, нужно каждый отрезок заменить на дугу (1/6 окружности с радиусом равным длине исходного отрезка). Получится более плавная фигура. После этого эту фигуру можно залить – теперь это терагон.
Это самый сложный фрактал из представленных сегодня. Программу попробуйте написать самостоятельно дома. За его выполнение получите дополнительный бал.

Фрактальны не только геометрические фигуры, фракталы встречаются и в природе. Самоподобие характерно для многих растений. Давайте посмотрим на примеры. (11-16 слайды)
8. Веточка (17 слайд)
Инициатором будет вертикальная линия размерностью 1. На следующей итерации рисуем уже 8 линий (длиной 0.5) следующим образом: 1ая линия рисуется из нижней точки исходной линии вверх, (хотя мы ее и нарисовали, но мы ее не увидим, потому что она наложится на исходную линию), следующие 7 рисуются из верхней вершины исходной линии в виде веера под углом 15( между собой. На следующей итерации для каждой из полученных линий рисуют свои 8 линий.

procedure draw(x,y,p,a:real;k:integer);

var p0,x1,y1:real;

    i,c:integer;

begin

  if (k<=0) or (p<1) then exit;

  k:=k-1;


//количество итераций и толщина линии
  p0:=p/2;

  c:=round(150*(1-k/4))+50;


//для игры с цветом
  c:=(c shl 16) or (c shl 8) or c;

//от черного до белого
  draw(x,y,p0,a,k);
  x1:=x+p*cos(a);

  y1:=y+p*sin(a);

  for i:=-3 to 3 do draw(x1,y1,p0,a+i*5* pi/36,k);

  form1.canvas.Pen.Color:=c;

  form1.Canvas.Pen.Width:=k;

  line(x,y,x1,y1);

end;

9. Укроп (18 слайд)
Инициатором для веточки укропа будет вертикальный отрезок длиной 1. На следующем шаге к нему пририсовывают еще 3 отрезка с длинами l1, l2, l3 меньшими 1, расположенными под углами a1, a2, a3 к основной линии. На следующей итерации такую же процедуру проделывают с этими 3 отрезками (к каждому пририсовывают по 3 отрезка) и т.д.

procedure draw(x,y,p,a:real;k:integer);

var x1,y1:real;

begin

  if (k=0) or (p<=1) then exit;

  k:=k-1;

  x1:=x+p*cos(a);

  y1:=y+p*sin(a);

  line0(x,y,x1,y1);

  draw(x1,y1,0.55*p,a+pi/6,k);
//коэффициенты выбираем сами
  x1:=x+0.6*p*cos(a);

  y1:=y+0.6*p*sin(a);

  draw(x1,y1,0.5*p,a+7*pi/36,k);

  x1:=x+0.4*p*cos(a);

  y1:=y+0.4*p*sin(a);

  draw(x1,y1,0.6*p,a-7*pi/36,k);

end;
10. Папоротник (19 слайд)
Найдите закономерность сами. Это будет вашим третьим домашним заданием. Правильность проверим на следующей паре.

Таким образом, что мы сегодня узнали на занятии? Что такое фрактал? Какие виды фракталов вы знаете? С какими геометрическими фракталами мы познакомились подробно? (отвечают на вопросы)
Ваше домашнее задание:
1) реализовать программу для построения салфетки Серпинского;
2) создать математическую модель для рисования папоротника;
3) терагон (необязательное задание).
PAGE  
11

_1350647291.unknown

